# NAG Library Routine Document

# G01EDF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

G01EDF returns the probability for the lower or upper tail of the $F$ or variance-ratio distribution with real degrees of freedom, via the routine name.

## 2 Specification

***double precision*** `FUNCTION G01EDF(TAIL, F, DF1, DF2, IFAIL)`

```
INTEGER                     IFAIL
double precision            F, DF1, DF2
CHARACTER*1                 TAIL
```

## 3 Description

The lower tail probability for the $F$, or variance-ratio distribution, with $\nu_1$ and $\nu_2$ degrees of freedom, $P(F \leq f : \nu_1, \nu_2)$, is defined by:

$$P(F \leq f : \nu_1, \nu_2) = \frac{\nu_1^{\nu_1/2}\nu_2^{\nu_2/2}\Gamma((\nu_1+\nu_2)/2)}{\Gamma(\nu_1/2)\Gamma(\nu_2/2)}\int_0^f F^{(\nu_1-2)/2}(\nu_1 F+\nu_2)^{-(\nu_1+\nu_2)/2}dF,$$

for $\nu_1,\ \nu_2 > 0,\ f \geq 0$.

The probability is computed by means of a transformation to a beta distribution, $P_\beta(B \leq \beta : a, b)$:

$$P(F \leq f : \nu_1, \nu_2) = P_\beta\left(B \leq \frac{\nu_1 f}{\nu_1 f + \nu_2} : \nu_1/2, \nu_2/2\right)$$

and using a call to G01EEF.

For very large values of both $\nu_1$ and $\nu_2$, greater than $10^5$, a normal approximation is used. If only one of $\nu_1$ or $\nu_2$ is greater than $10^5$ then a $\chi^2$ approximation is used, see Abramowitz and Stegun (1972).

## 4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

## 5 Parameters

1: TAIL – CHARACTER*1 *Input*

*On entry*: indicates whether an upper or lower tail probability is required.

TAIL = 'L'
The lower tail probability is returned, i.e., $P(F \leq f : \nu_1, \nu_2)$.

TAIL = 'U'
The upper tail probability is returned, i.e., $P(F \geq f : \nu_1, \nu_2)$.

*Constraint*: TAIL = 'L' or 'U'.

2: F – ***double precision*** *Input*

*On entry*: $f$, the value of the $F$ variate.

*Constraint*: F $\geq 0$.

3: DF1 – ***double precision*** *Input*

*On entry*: the degrees of freedom of the numerator variance, $\nu_1$.

*Constraint*: DF1 $> 0.0$.

4: DF2 – ***double precision*** *Input*

*On entry*: the degrees of freedom of the denominator variance, $\nu_2$.

*Constraint*: DF2 $> 0.0$.

5: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to $0$, $-1$ or $1$. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

*On exit*: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or $1$ is recommended. If the output of error messages is undesirable, then the value $1$ is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is $-1$. **When the value $-1$ or $1$ is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

**Note**: G01EDF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If IFAIL $= 1$, $2$ or $3$ on exit, then G01EDF returns $0.0$.

IFAIL $= 1$

On entry, TAIL $\neq$ 'L' or 'U'.

IFAIL $= 2$

On entry, F $< 0.0$.

IFAIL $= 3$

On entry, DF1 $\leq 0.0$,
or DF2 $\leq 0.0$.

IFAIL $= 4$

F is too far out into the tails for the probability to be evaluated exactly. The result tends to approach $1.0$ if $x$ is large, or $0.0$ if $x$ is small. The result returned is a good approximation to the required solution.

## 7 Accuracy

The result should be accurate to five significant digits.

# 8 Further Comments

For higher accuracy G01EEF can be used along with the transformations given in Section 3.

# 9 Example

This example reads values from, and degrees of freedom for F-distributions, computes the lower-tail probabilities and prints all these values until the end of data is reached.

## 9.1 Program Text

```
*     G01EDF Example Program Text
*     Mark 14 Release. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION DF1, DF2, F, PROB
      INTEGER          IFAIL
*     .. External Functions ..
      DOUBLE PRECISION G01EDF
      EXTERNAL         G01EDF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G01EDF Example Program Results'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '   F      DF1     DF2    PROB'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
   20 READ (NIN,*,END=40) F, DF1, DF2
      IFAIL = 1
*
      PROB = G01EDF('Lower',F,DF1,DF2,IFAIL)
*
      IF (IFAIL.GE.0) THEN
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,99999) F, DF1, DF2, PROB
         ELSE
            WRITE (NOUT,99999) F, DF1, DF2, PROB, ' NOTE: IFAIL = ',
     +        IFAIL
         END IF
         GO TO 20
      ELSE
         WRITE (NOUT,99998) ' ** G01EDF returned with IFAIL = ', IFAIL
      END IF
   40 CONTINUE
*
99999 FORMAT (1X,F6.3,2F8.3,F8.4,A,I1)
99998 FORMAT (1X,A,I5)
      END
```

## 9.2 Program Data

```
G01EDF Example Program Data
  5.5   1.5   25.5
 39.9   1.0    1.0
 2.5   20.25   1.0
```

## 9.3 Program Results

```
 G01EDF Example Program Results

    F       DF1      DF2     PROB

  5.500    1.500   25.500   0.9837
 39.900    1.000    1.000   0.9000
  2.500   20.250    1.000   0.5342
```